

Excel. Extraer datos de una celda

En ocasiones necesitamos extraer contenido de una celda de Excel para utilizarlo en alguna otra fórmula o simplemente porque simplemente necesitamos una parte de la cadena de texto contenida en la celda.

Extraer caracteres al inicio o al final

Si necesitas extraer un número determinado de caracteres que se encuentran al inicio o al final de una cadena de texto podemos utilizar la función IZQUIERDA o la función DERECHA.

Observa el siguiente ejemplo que extrae los primeros 5 caracteres de la cadena de texto en A1:

C3		fx =IZQUIERDA(A1, 5)		
	A	B	C	D
1	Contenido de una celda			
2				
3		Primeros 5 caracteres:	Conte	
4				

En este ejemplo utilice la función IZQUIERDA pero puedo utilizar la función DERECHA para extraer los últimos 4 caracteres de la misma cadena:

C4		fx =DERECHA(A1, 4)		
	A	B	C	D
1	Contenido de una celda			
2				
3		Primeros 5 caracteres:	Conte	
4		Últimos 4 caracteres:	elda	
5				

Extraer caracteres intermedios

Si por el contrario, necesitas extraer sólo una parte que se encuentra en medio de la cadena de texto, podemos utilizar la función EXTRAER que nos permite indicar la posición inicial dentro de la cadena donde se iniciará la extracción y adicionalmente la cantidad de caracteres que deseamos extraer.

En el siguiente ejemplo podrás observar que he extraído la palabra “una” especificando a la función EXTRAER que inicie en la posición 14 y obtenga solamente 3 caracteres.

C5		fx =EXTRAE(A1, 14, 3)		
	A	B	C	D
1	Contenido de una celda			
2				
3		Primeros 5 caracteres:	Conte	
4		Últimos 4 caracteres:	elda	
5		En medio de cadena:	una	
6				

Para poder utilizar la función EXTRAE es indispensable conocer la posición inicial donde comenzará la extracción y su longitud en caracteres.

Las tres funciones mencionadas: IZQUIERDA, DERECHA, EXTRAE son las funciones con las cuales podemos **extraer contenido de una celda en Excel** y a partir de estas funciones básicas podemos hacer combinaciones con otras funciones para otros usos como los que menciono a continuación.

Extraer la primera palabra

Para extraer la primera palabra de una celda utilizamos la función IZQUIERDA junto con la función ENCONTRAR que nos permitirá encontrar la posición del primer espacio en blanco lo cual delimita la primera palabra. La fórmula utilizada es la siguiente:

```
=IZQUIERDA(A1, ENCONTRAR(" ", A1) - 1)
```

El primer argumento de la función ENCONTRAR es un espacio en blanco y al resultado devuelto le resta el número 1 para poder obtener la longitud correcta de caracteres de la primera palabra. Observa el resultado:

C6		fx =IZQUIERDA(A1, ENCONTRAR(" ", A1) - 1)		
	A	B	C	D
1	Contenido de una celda			
2				
3		Primeros 5 caracteres:	Conte	
4		Últimos 4 caracteres:	elda	
5		En medio de cadena:	una	
6		Primera palabra:	Contenido	
7				

Extraer la última palabra

Para poder extraer la última palabra la situación se complica porque es necesario encontrar el último espacio y a partir de ahí tomar los caracteres de la derecha. Pero para simplificarte el trabajo, esta es la fórmula que obtiene el resultado adecuado:

```
=ESPACIOS(DERECHA(SUSTITUIR(A1, " ", REPETIR(" ", LARGO(A1))), LARGO(A1)))
```

Esta fórmula devolverá el resultado correcto en nuestro ejemplo:

C7		fx =ESPACIOS(DERECHA(SUSTITUIR(A1," ", REPETIR(" ",LARGO(A1))),LARGO(A1)))			
	A	B	C	D	E
1	Contenido de una celda				
2					
3		Primeros 5 caracteres:	Conte		
4		Últimos 4 caracteres:	elda		
5		En medio de cadena:	una		
6		Primera palabra:	Contenido		
7		Última palabra:	celda		
8					

Extraer la n-ésima palabra

Si lo que necesitas es extraer algún número de palabra específica puedes utilizar la siguiente fórmula:

=ESPACIOS(EXTRAE(SUSTITUIR(A1," ",REPETIR(" ",LARGO(A1))), (D8-1)*LARGO(A1)+1, LARGO(A1)))

En esta fórmula el número de palabra se encuentra en la celda D8 por lo que si quiero obtener la segunda palabra debo colocar el valor 2 en dicha celda. Observa el ejemplo:

C8		fx =ESPACIOS(EXTRAE(SUSTITUIR(A1," ",REPETIR(" ",LARGO(A1))), (D8-1)*LARGO(A1)+1, LARGO(A1)))				
	A	B	C	D	E	F
1	Contenido de una celda					
2						
3		Primeros 5 caracteres:	Conte			
4		Últimos 4 caracteres:	elda			
5		En medio de cadena:	una			
6		Primera palabra:	Contenido			
7		Última palabra:	celda			
8		La palabra n:	de	2		
9						